

Creating Drill Templates using METAPOST

(and a Snowman too)

John C. Luciani Jr.

December 1, 2005

Overview

For prototyping, paper templates provide a quick and accurate way to perform drilling, routing or cutting operations. The METAPOST picture-drawing language provides an easy way to create these templates. METAPOST implements a picture-drawing language similar to Knuth's METAFONT except that the output is PostScript.

Heatsink Drill Template

The drill template in figure 2 is used to drill and tap holes for mounting five Lumiled Luxeon Star LEDs on a heatsink.

Lines 14-30 define a function that draws an LED outline and two mounting holes. To draw the outline a square path is created and then drawn by shifting the lower-left corner to the appropriate coordinates.

Lines 32-41 define a function to draw a mounting hole outline with crosshairs. A path is defined for each of the crosshair lines. The mounting hole outline and crosshairs are drawn with a line thickness of 1pt.

Lines 44-64 draw the heatsink template. The point $z0$ is used to offset the template drawing on the output page.

Lines 52-55 output a top and bottom outline. $p1$ is the path containing the bottom outline. To get the top outline $p1$ is reflected about the horizontal centerline of the heatsink.

Lines 59-62 equally space the LEDs across the heatsink.




Figure 1: Heatsink Template for Mounting Five Luxeon Star/C LEDs

Listing 1: Heatsink Template

```

1 L = 25mm;      % length of Luxeon/C (x direction)
2 W = 25mm;      % width of Luxeon/C (y direction)
3
4 % 4-40 thread requires a drill size of #43 (0.089)
5 % for soft metals
6
7 d = 0.089in;   % diameter of drill bit
8 offset=3.5mm;  % distance from the edge of the Luxeon/C to the mounting hole
9 HS_length=5.34375in; % length of the heatsink (5 11/32")
10 HS_width=2in;  % width of the heatsink
11
12 % (x, y) is the lower-left corner of the LED
13
14 def draw_led(expr x, y) =
15     save p;
16     path p[];
17     p1 = (0,0)--(0,L)--(L,L)--(L,0)--cycle; % body of the Luxeon/C
18
19     % draw the Luxeon/C body with line thickness of 2pt
20
21     pickup pencircle scaled 2pt;
22     draw p1 shifted (x,y);
23
24     % draw the mounting holes, upper left and lower right, and
25     % crosshairs with a line thickness of 1pt.
26
27     draw_mounting_hole(x+offset,y+W-offset); % upper left
28     draw_mounting_hole(x+L-offset,y+offset); % lower right
29
30 enddef;
31
32 def draw_mounting_hole(expr x,y) =
33     save p;
34     path p[];
35     p1 = (-d/2,0)--(d/2,0); % horizontal line across the drill hole diameter
36     p2 = (0,-d/2)--(0,d/2); % vertical line across the drill hole diameter
37     pickup pencircle scaled 1pt;
38     draw fullcircle scaled d shifted (x,y);
39     draw p1 shifted (x,y);
40     draw p2 shifted (x,y);
41 enddef;
42
43
44 beginfig(1);
45     path p[];
46
47     z0 = (1in,1in); % offset for the laserjet
48
49     % Draw an outline to aid in positioning the template on the
50     % heatsink.
51
52     pickup pencircle scaled 2pt;
53     p1 = (0,0.25in) -- (0,0) -- (HS_length,0) -- (HS_length,0.25in);
54     draw p1 shifted z0; % bottom outline
55     draw p1 reflectedabout ((0,1in),(1in,1in)) shifted z0; % top outline
56
57     % Place five LEDs, equally spaced, on the heatsink.
58
59     spacing = (HS_length - 5L)/6;
60     for i=1 upto 5:
61         draw_led(i*spacing + (i-1)*L + x0, HS_width/2 - W/2 + y0);
62     endfor
63
64 endfig;
65
66
67
68
69 end.
```

Snowman




Figure 2: Snowman

Listing 2: Snowman

```
1 penwidth = 4pt;
2 pageoffset = 0.5in; % margin for the laserjet
3
4 def draw_snowman(expr w) =
5   save a, b, c;
6
7   pickup pencircle scaled penwidth;
8
9   % circle centers
10
11  z0 = (w/2+pageoffset,w/2+pageoffset); % bottom circle
12  z1 = (x0,1.375w-w/2+pageoffset);     % middle circle
13  z2 = (x0,1.875w-w/2+pageoffset);     % top circle
14
15  % circle diameters
16
17  a = w;           % bottom circle
18  b = 3w/4;       % middle circle
19  c = 0.5625w;    % top circle
20
21  % draw the circles and then erase the inner lines
```

```

22
23 draw fullcircle scaled a shifted z0;
24 draw fullcircle scaled b shifted z1;
25 draw fullcircle scaled c shifted z2;
26
27 unfilldraw fullcircle scaled (a - 2penwidth) shifted z0;
28 unfilldraw fullcircle scaled (b - 2penwidth) shifted z1;
29 unfilldraw fullcircle scaled (c - 2penwidth) shifted z2;
30
31 % eyes
32
33 for i=-1 step 2 until 1:
34     filldraw fullcircle scaled 0.1c shifted (x2 + (3/16)*i*x2, y2 + c/4);
35 endfor;
36
37 % nose
38
39 filldraw fullcircle scaled 0.2c shifted z2;
40
41 % mouth
42
43 path c[];
44 c[0] = fullcircle scaled 0.1c shifted (x2, y2-c/4); % 7.85
45 for i=50 step -25 until -50:
46     filldraw c[0] rotatedaround(z2, i);
47 endfor;
48
49
50 enddef;
51
52 beginfig(1); draw_snowman(2in); endfig;
53
54 end.

```

References

Hobby, J. D. (1992). *A User's Manual for MetaPost* (Tech. Rep. No. no. 162).
Knuth, D. E. (1986). *The METAFONTbook* (Vol. C). Reading, MA, USA: Addison-Wesley.

License

No-Fee License Version 0.1

Intent

The intent of this license is to allow for distribution of this documentation without fee. Usage of this documentation, other than distribution, is unrestricted.

License

This documentation may be distributed provided that no fee is charged. Modification is allowed provided that the resultant work is released with the same license.

This documentation is provided by the author **“AS IS”** and any express or implied warranties, including, but not limited to, the implied warranties of merchantability and fitness for a particular purpose are disclaimed. In no event shall the author be liable for any direct, indirect, incidental, special, exemplary, or consequential damages (including, but not limited to, procurement of substitute goods or services; loss of use, data, or profits; or business interruption) however caused and on any theory of liability, whether in contract, strict liability, or tort (including negligence or otherwise) arising in any way out of the use of this software, even if advised of the possibility of such damage.